

EXCEL

Support de cours


Présentation de l'interface

La feuille de calcul


Une feuille de calcul comprend :

- 16 384 (2^{14}) colonnes, numérotées de A à XFD ;
- 1 048 576 (2^{20}) lignes, numérotées de 1 à 1 048 576.

Soit environ 17 milliards de cellules (17 179 869 184 exactement), de la cellule A1 à la cellule XFD1048576.



Lorsqu'on travaille sur de grandes feuilles de calcul, il est possible d'en « fractionner » l'affichage ; chaque partie de la feuille s'affiche alors dans un volet indépendant :




Notion de classeur

Un même fichier Excel peut comprendre plusieurs feuilles de calcul réunies dans un même « classeur ». Les contrôles permettant de gérer les feuilles sont regroupés en bas à gauche. Un bouton permet de créer une nouvelle feuille. Les feuilles (nommées Feuil1, Feuil2, Feuil3, etc. par défaut) peuvent être renommées (clic secondaire). Les feuilles sont accessibles au moyen d'un onglet. On peut réorganiser les feuilles en faisant glisser les onglets. Lorsqu'un classeur comprend un grand nombre de feuilles, on peut faire défiler les onglets au moyen des « boutons de magnéscope ».

Le « ruban »

Depuis Office 2007, un « ruban » remplace les barres d'outils et les menus ; les fonctions les plus utiles sont regroupées dans des rubans (Accueil, Insertion, Mise en page, Formules, Données, Révision, Affichage et Compléments).




Utilisation de la souris


Sélectionner une cellule : cliquer au centre de la cellule ; glisser pour étendre la sélection.




Sélectionner une colonne : cliquer sur l'en-tête de la colonne ; glisser pour étendre la sélection.




Sélectionner une ligne : cliquer sur l'en-tête de la ligne ; glisser pour étendre la sélection.




Sélectionner tout : cliquer à l'intersection des en-têtes de lignes et de colonne.





Recopier une cellule : cliquer sur le point de recopie et tirer vers le bas ou vers la droite. Maintenir la touche Ctrl appuyée permet de recopier avec ou sans incrémentation des valeurs. Double-clic sur le point de recopie pour recopier vers le bas dans un tableau déjà rempli.





Déplacer une cellule ou une plage de cellules : sélectionner la cellule ou la plage de cellules, cliquer sur le contour de la sélection et glisser.



Redimensionner les colonnes ou les lignes : cliquer entre les colonnes ou les lignes (dans les zones d'en-têtes) et tirer ; double-clic pour ajuster au contenu.



Insérer une ligne ou une colonne : sélectionner la ligne ou la colonne, clic secondaire et Insertion. Sélectionner plusieurs lignes ou colonnes pour insérer plusieurs lignes ou colonnes.



Utilisation du clavier

Se déplacer dans le tableau

- Ctrl + → Dernière colonne
- Ctrl + ← Première colonne
- Ctrl + ↓ Dernière ligne
- Ctrl + ↑ Première ligne
- Ctrl + Début Première cellule
- Ctrl + Fin Dernière cellule active

Touches d'édition

- Ctrl + C Copier
- Ctrl + X Couper
- Ctrl + V Coller
- Ctrl + Z Annuler

Recopier

- Ctrl + B Recopier vers le bas
- Ctrl + D Recopier vers la droite

Sélectionner

Ctrl + **Espace** Colonne courante

Maj + **Espace** Ligne courante

Ctrl + **A** Sélectionner tout

Étendre la sélection

Maj + Clic

Maj +    

Ctrl + **Maj** +    

Ajouter à la sélection

Ctrl + Clic

Touches de fonction

Entrée Valider

F1 Aide


Maj + **F2** Ajouter / Modifier un commentaire

Maj + **F3** Insérer une fonction

F4 Répéter l'action précédente

F5 Afficher la boîte de dialogue *Atteindre*

Maj + **F5** Afficher la boîte de dialogue *Rechercher et remplacer*

Ctrl + **Maj** +  Afficher la boîte de dialogue *Format de cellule*

Premier tableau

Soit à réaliser le tableau suivant :

	A	B	C	D
1	Désignation	PU	Qté	Montant HT
2	Bureau	250,00 €	1	250,00 €
3	Fauteuil	150,00 €	1	150,00 €
4	Lampe	50,00 €	1	50,00 €
5	Téléphone	65,00 €	1	65,00 €
6	Ordinateur	850,00 €	1	850,00 €
7	Imprimante	450,00 €	1	450,00 €
8	Ramette papier 90 g	15,00 €	5	75,00 €
9	Cartouche d'encre	55,00 €	2	110,00 €
10	Clé USB	22,00 €	2	44,00 €
11	Agenda	18,00 €	1	18,00 €
12	Stylo	1,00 €	5	5,00 €
13	Gomme	1,50 €	1	1,50 €
14	Ciseaux	2,50 €	1	2,50 €
15	Colle	0,75 €	3	2,25 €
16	Ruban adhésif	0,95 €	3	2,85 €
17				
18	Total HT			2 076,10 €
19	TVA		19,60%	406,92 €
20	Montant TTC			2 483,02 €

Montant HT

En D2, entrer la formule : =B2*C2 et recopier vers le bas (de D2 à D16).

Vous pouvez utiliser les opérateurs arithmétiques « + » (plus), « - » (moins), « * » (multiplié par) et « / » (divisé par) et le symbole « ^ » (élévation à la puissance) dans les formules.

Total HT

En D18, entrer la formule : =SOMME(D2:D16).

SOMME est une fonction intégrée ; pour insérer une fonction :

- à gauche de la barre de formule : cliquer sur f_x (insérer une fonction), sélectionner la fonction, et entrer les arguments de cette fonction ;
- ou bien : taper « = » et sélectionner la fonction dans la liste déroulante *Fonctions*, à gauche de la barre de formule. La liste comprend les 10 dernières fonctions utilisées ; si la fonction souhaitée ne figure pas dans la liste sélectionnez *Autres fonctions*...
- ou bien : taper directement le nom de la fonction dans la barre de formule en respectant la syntaxe indiquée.


TVA et Montant TTC

En D19 (TVA) entrer la formule =D18*C19.

En D20 (Montant TTC), entrer la formule = D18+D19.


Le formatage des cellules

Cette boîte de dialogue donne accès à toutes les options de formatage des cellules (formats de nombres, alignement du contenu, polices de caractères, bordures, remplissage) ainsi qu'aux options de protection des cellules.



La mise en page et le contrôle de l'impression

Le ruban *Mise en page* donne accès aux principales fonctions de mise en page et de contrôle de l'impression, ainsi qu'à la boîte de dialogue *Mise en page*. On y trouvera des options spécifiques permettant notamment d'ajuster automatiquement les dimensions du tableau à celle de la feuille, et de centrer le tableau dans la page. Pour les tableaux de grandes dimensions, il sera possible de définir l'ordre d'impression des pages, ainsi que, le cas échéant, les lignes à répéter en haut et/ou les colonnes à répéter à gauche.



Dans le ruban *Mise en page*, des boutons permettent de définir la zone d'impression et d'insérer manuellement des sauts de page aux lignes et/ou colonnes souhaitées.

Le système de références

Références absolues

Par défaut les références aux cellules sont notées sous la forme de références *absolues* (A1, B1, C1... A2, B2, C2, etc.) :

	A	B	C	D
1	Désignation	PU	Qté	Montant
2	Pomme	1,95 €	1	=B2*C2
3	Poire	2,35 €	2	=B3*C3
4	Abricot	3,25 €	1	=B4*C4
5				
6	Total		=SOMME(C2:C4)	=SOMME(D2:D4)

N.B. : Lorsqu'on recopie vers le bas ou vers la droite, Excel incrémente automatiquement les numéros de ligne et les lettres des colonnes dans les références des cellules.

Références relatives

Pour afficher les références aux cellules sous la forme de références *relatives*, il faut passer en mode d'affichage « L1C1 » :

Fichier : Options : Formules : Manipulation de formules. Cocher la case *Style de référence L1C1*.

On obtient alors :

	A	B	C	D
1	Désignation	PU	Qté	Montant
2	Pomme	1,95 €	1	=LC(-2)*LC(-1)
3	Poire	2,35 €	2	=LC(-2)*LC(-1)
4	Abricot	3,25 €	1	=LC(-2)*LC(-1)
5				
6	Total		=SOMME(L(-4)C:L(-2)C)	=SOMME(L(-4)C:L(-2)C)

Figurer la référence à une ligne ou à une colonne lors de la recopie de formules

Pour figer la référence à une ligne (lors de la recopie vers le bas) ou à une colonne (lors de la recopie vers la droite), il faut insérer le caractère « \$ » devant le numéro de la ligne ou la lettre de la colonne. Pour figer la référence à la fois à une ligne et à une colonne (lors de la recopie vers le bas et vers la droite), il faut insérer le caractère « \$ » devant le numéro de la ligne et de la colonne.

	A	B	C
1	Propriétaire	Surface	Charges
2	Pierre Dupont	30 m ²	3 000,00 €
3	Pauline Carton	90 m ²	9 000,00 €
4	Jean Aymar	60 m ²	6 000,00 €
5			
6	Montant total des charges à répartir		18 000,00 €

	C
	Charges
	=C\$6/SOMME(B\$2:B\$4)*C2
	=C\$6/SOMME(B\$2:B\$4)*C3
	=C\$6/SOMME(B\$2:B\$4)*C4

Astuce : appuyer une ou plusieurs fois sur la touche F4 pour placer le caractère « \$ » devant le numéro de ligne ou la lettre de la colonne :

C6 [F4] \$C\$6 [F4] C\$6 [F4] \$C2 [F4] C6

En mode d'affichage L1C1, on obtiendrait :

	A	B	C
1	Propriétaire	Surface	Charges
2	Pierre Dupont	30 m ²	3 000,00 €
3	Pauline Carton	90 m ²	9 000,00 €
4	Jean Aymar	60 m ²	6 000,00 €
5			
6	Montant total des charges à répartir		35 000,00 €

	C
	Charges
	=L6C/SOMME(L2C(-1):L4C(-1))*LC(-1)
	=L6C/SOMME(L2C(-1):L4C(-1))*LC(-1)
	=L6C/SOMME(L2C(-1):L4C(-1))*LC(-1)

Les fonctions

La fonction Arrondi

Formules

	A	B	C
1		Montant HT	TVA
2		6	=B2*19,6%
3		5,4	=B3*19,6%
4	Total	11,4	=SOMME(C2:C3)

Résultat

C
TVA
1,176
1,0584
2,2344

Appliquons un format monétaire

	A	B	C
1		Montant HT	TVA
2		6,00 €	1,18 €
3		5,40 €	1,06 €
4	Total	11,40 €	2,23 €

On obtient : **1,18 + 1,06 = 2,23** (au lieu de **2,24** !)

Pour arrondir les résultats à deux décimales, utiliser la fonction ARRONDI !

Formules

	A	B	C
1		Montant HT	TVA
2		6,00 €	=ARRONDI(B2*19,6%;2)
3		5,40 €	=ARRONDI(B3*19,6%;2)
4	Total	11,40 €	=SOMME(C2:C3)

Résultat

	A	B	C
1		Montant HT	TVA
2		6,00 €	1,18 €
3		5,40 €	1,06 €
4	Total	11,40 €	2,24 €

On obtient maintenant : **1,18 + 1,06 = 2,24**

Les fonctions statistiques

Les fonctions Somme, Moyenne, Nb et NbVal

SOMME Somme d'une série de valeurs numériques contenue dans une [ou plusieurs] plages de cellules.

MOYENNE Moyenne arithmétique d'une série de valeurs numériques contenue dans une [ou plusieurs] plages de cellules.

NB Nombre de cellules contenant une valeur numérique dans une [ou plusieurs] plages de cellules.

NBVAL Nombre de cellules non vides (pouvant contenir du texte ou une valeur numérique) dans une [ou plusieurs] plages de cellules.

Syntaxe : FONCTION (plage1 ; [plage2] ; [plage3] ; etc...).

Les formules suivantes sont équivalentes :

MOYENNE(plage) = SOMME(plage) / NB(plage)

Toujours utiliser la fonction MOYENNE pour effectuer un calcul de moyenne !

Les fonctions Minimum et Maximum

MIN Valeur minimum d'une série de valeurs numériques contenues dans une [ou plusieurs] plages de cellules.

MAX Valeur maximum d'une série de valeurs numériques contenues dans une [ou plusieurs] plages de cellules.

Les fonctions Moyenne, Variance et Écart type

L'écart type est un indicateur de dispersion ; il exprime l'écart type des valeurs d'une série par rapport à la moyenne de ces valeurs : plus il est faible, plus la population est homogène – et plus il est élevé, plus la population est hétérogène. L'écart type est défini comme étant égal à la racine carrée de la variance.

Définitions mathématiques :

Moyenne	Variance	Écart type
$m_x = \frac{\sum n_i \times x_i}{n}$	$V_x = \frac{\sum n_i \times x_i^2}{n} - m_x^2$	$\sigma_x = \sqrt{V_x}$

Application :

	A	B	C	D
1	Age x_i	Effectif n_i	$n_i \times x_i$	$n_i \times x_i^2$
2	17	1	17	289
3	18	3	54	972
4	19	3	57	1083
5	20	2	40	800
6	21	2	42	882
7				
8	Somme	11	210	4026
9				
10	Moyenne	19,09		
11	Variance	1,54		
12	Écart-type	1,24		

	A	B	C	D
1	Age x_i	Effectif n_i	$n_i \times x_i$	$n_i \times x_i^2$
2	17	1	=B3*A3	=B3*A3^2
3	18	3
4	19	3
5	20	2
6	21	2
7				
8	Somme	=SOMME(B2:B6)
9				
10	Moyenne	=C8/B8		
11	Variance	=D8/B8-B10^2		
12	Écart-type	=RACINE(B11)		

On dispose des fonctions intégrées VAR.P.N (variance) et ECARTYPE.PEARSON (écart type) ; soit la série suivante :

	A	B	C
1		x_i	x_i^2
2		15	225
3		8	64
4		16	256
5		5	25
6		13	169
7		18	324
8		8	64
9		12	144
10		12	144
11		8	64
12			
13	Moyenne	11,50	
14	Variance	15,65	
15	Écart type	3,96	

Les formules suivantes sont équivalentes :

Variance	Écart type
=MOYENNE(C2:C11)-MOYENNE(B2:B11)^2	=RACINE(B15)
=VAR.P.N(B2:B11)	=ECARTYPE.PEARSON(B2:B11)

Médiane, quartiles, déciles et centiles

La médiane, les quartiles, les déciles et les centiles sont des indicateurs de répartition de la population (répartition des effectifs).

La médiane (à ne pas confondre avec la moyenne) est la valeur qui sépare la population en deux :

- celle qui est au-dessus de cette valeur (50% des effectifs) ;
- et celle qui est au-dessous (50% des effectifs).

Une population se répartit également en :

- quatre quartiles comprenant chacun 25% de la population ;
- dix déciles comprenant chacun 10% de la population ;
- cent centiles comprenant chacun 1% de la population ;

Soit une série de données, nous pouvons utiliser les fonctions suivantes :

Minimum		=MIN(série)
Maximum		=MAX(série)
Médiane		=MEDIANE(série)
Minimum		=QUARTILE(série;0)
1 ^{er} quartile		=QUARTILE(série;1)
Médiane	(2 ^e quartile)	=QUARTILE(série;2)
3 ^e quartile		=QUARTILE(série;3)
Maximum		=QUARTILE(série;4)
Minimum		=CENTILE(série;0%)
1 ^{er} centile		=CENTILE(série;1%)
1 ^{er} décile	(10 ^e centile)	=CENTILE(série;10%)
1 ^{er} quartile	(25 ^e centile)	=CENTILE(série;25%)
Médiane	(50 ^e centile)	=CENTILE(série;50%)
3 ^e quartile	(75 ^e centile)	=CENTILE(série;75%)
9 ^e décile	(90 ^e centile)	=CENTILE(série;90%)
99 ^e centile		=CENTILE(série;99%)
Maximum		=CENTILE(série;100%)

Les fonctions logiques

Excel reconnaît les valeurs logiques VRAI et FAUX. On dispose également des opérateurs logiques ET, OU et NON sous la forme de fonctions.

Soient deux propositions P_1 et P_2 , nous pouvons réaliser la table de vérité suivante :

	A	B	C	D	E
1	P_1	P_2	NON P_1	P_1 ET P_2	P_1 OU P_2
2	FAUX	FAUX	=NON(A2)	=ET(A2;B2)	=OU(A2;B2)
3	FAUX	VRAI	=NON(A3)	=ET(A3;B2)	=OU(A3;B2)
4	VRAI	FAUX	=NON(A4)	=ET(A4;B2)	=OU(A4;B2)
5	VRAI	VRAI	=NON(A5)	=ET(A5;B2)	=OU(A5;B2)

On obtient le résultat suivant :

	A	B	C	D	E
1	P_1	P_2	NON P_1	P_1 ET P_2	P_1 OU P_2
2	FAUX	FAUX	VRAI	FAUX	FAUX
3	FAUX	VRAI	VRAI	FAUX	VRAI
4	VRAI	FAUX	FAUX	FAUX	VRAI
5	VRAI	VRAI	FAUX	VRAI	VRAI

Les fonctions permettant de tester le type de contenu des cellules

La fonction ESTVIDE permet de savoir si une cellule est vide ou non.

La fonction : permet de savoir si une cellule contient :

ESTTEXTE une chaîne de caractères

ESTNONTEXTE une autre valeur qu'une chaîne de caractères

ESTNUM une valeur numérique

ESTLOGIQUE une valeur logique

Ces fonctions renvoient la valeur VRAI ou FAUX.

La fonction TYPE renvoie un nombre correspondant au type de données contenu dans la cellule :

1 pour une valeur numérique

2 pour une chaîne de caractères

3 pour une valeur logique

4 pour une erreur

5 pour une matrice

La gestion des erreurs

En cas d'erreur dans une formule ou d'erreur de calcul, Excel renvoie un code d'erreur dans la cellule :

N°	Exemple	Erreur
1	=SOMME(\$A\$1 \$A\$2)	#NUL!
2	=1/0	#DIV/0!
3	=1+"Toto"	#VALEUR!
4	=#REF!	#REF!
5	=Toto	#NOM?
6	=RACINE(-1)	#NOMBRE!
7	=RECHERCHEV("Toto";\$A\$1;1;FAUX)	#N/A

La fonction TYPE.ERREUR renvoie un nombre correspondant à un numéro d'erreur (ou #N/A s'il n'y a pas d'erreur). La fonction ESTERREUR renvoie la valeur VRAI en cas d'erreur. La fonction ESTERR renvoie la valeur VRAI en cas d'erreur, sauf pour #N/A. La fonction ESTNA renvoie la valeur VRAI en cas d'erreur #N/A.

La fonction SIERREUR permet d'afficher un message personnalisé en cas d'erreur.

La fonction SI

On utilise la fonction SI lorsqu'un résultat est soumis à une ou plusieurs conditions.

Syntaxe : SI (condition ; [valeur si vrai] ; [valeur si faux])

Les fonctions SI peuvent être imbriquées pour exprimer des conditions complexes.

Exemple simple

On accorde une remise de 5% pour toute commande d'un montant supérieur ou égal à 500€.

	A	B	C
1	Montant avant remise	Remise	Montant après remise
2	1 057,85 €	52,89 €	1 004,96 €
3	609,51 €	0,00 €	609,51 €
4	1 434,12 €	71,71 €	1 362,41 €

Expression de la règle au moyen d'un pseudo-langage

SI Montant < 1000 **ALORS**
Pas de remise
SINON
Remise = Montant x 5%
FIN SI

Formule Excel

Calcul du montant de la remise (cellule B2) :

=SI(A2<1000;0;A2*5%)

On obtient 0 lorsqu'on le montant de la commande est inférieur à 1 000 € ; pour ne pas afficher les valeurs 0 dans une feuille de calcul :

Fichier : Options : Options Excel : Options avancées : Afficher les options pour cette feuille de calcul

Ne pas cocher la case de l'option « Afficher un zéro dans les cellules qui ont une valeur nulle ».

On aurait aussi pu utiliser la formule suivante :

=SI(A2<1000;"";A2*5%)

Dans ce cas, on obtient "" (chaîne de caractères vide) lorsqu'on le montant de la commande est inférieur à 1 000 €. Mais on obtient un message d'erreur « #VALEUR! » pour le calcul du montant après remise ; en effet, il n'est pas possible de soustraire une chaîne de caractères vides à un nombre ; pour éviter cette erreur, la formule à utiliser pour le calcul du montant après remise (cellule C2) est :

=SI(B2="" ; A2 ; A2-B2)

Pour éviter l'affichage d'un zéro lorsque le montant avant remise n'est pas renseigné, il suffit de faire un test ; la formule devient :

=SI(ESTVIDE(A2) ; "" ; SI(B2="" ; A2 ; A2-B2))

Exemple avec fonctions SI imbriquées

On accorde une remise dans les conditions suivantes :

Montant de la commande	Montant de la remise
Commande < 1000 €	Pas de remise
1000 <= Commande < 2000	Remise de 5%
2000 <= Commande	Remise de 10%

Expression de la règle au moyen d'un pseudo-langage

SI Montant < 1000 **ALORS**
Pas de remise
SINON
 SI Montant < 2000 **ALORS**
 Remise = Montant x 5%
 SINON
 Remise = Montant x 10%
 FIN SI
FIN SI

Formule Excel

Calcul du montant de la remise (cellule B2) :

=SI(A2<1000;0;SI(A2<2000;A2*5%;A2*10%))

Mode opératoire :

Arguments de la fonction

SI

Test_logique A2<1000 = FAUX

Valeur_si_vrai 0 = 0

Valeur_si_faux = quelconque = FAUX

Vérifie si la condition est respectée et renvoie une valeur si le résultat d'une condition que vous avez spécifiée est VRAI, et une autre valeur si le résultat est FAUX.

Valeur_si_faux représente la valeur renvoyée si test logique est FAUX. Si omis, FAUX est renvoyé.

Résultat = FAUX

Aide sur cette fonction

OK Annuler

Test_logique A2<1000

Valeur_si_vrai 0

Valeur_si_faux Le curseur étant dans cette zone, cliquer sur SI en haut à gauche pour imbriquer la 2^e fonction SI.

Arguments de la fonction

SI

Test_logique A2<2000 = VRAI

Valeur_si_vrai A2*5% = 52,8925

Valeur_si_faux A2*10% = 105,785

Vérifie si la condition est respectée et renvoie une valeur si le résultat d'une condition que vous avez spécifiée est VRAI, et une autre valeur si le résultat est FAUX.

Valeur_si_faux représente la valeur renvoyée si test logique est FAUX. Si omis, FAUX est renvoyé.

Résultat = 52,89 €

Aide sur cette fonction

OK Annuler

Test_logique A2<2000

Valeur_si_vrai A2*5%

Valeur_si_faux A2*10%

Cliquer sur OK

N.B. : Observez la progression de la saisie dans la zone de formule.

On peut imbriquer autant de fonctions SI que nécessaire, selon le nombre de conditions :

- une fonction SI permet de traiter deux cas
- deux fonctions SI permettent de traiter trois cas
- trois fonctions SI permettent de traiter quatre cas
- etc.

Tableau de paramètres

Les conditions de remise (seuils et taux) pouvant varier, il est vivement conseillé de créer un tableau de paramètres, comme ci-dessous :

	A	B	C
1	Montant de la commande	Seuil	Taux de remise
2	– inférieur à	1 000 €	0%
3	– inférieur à	2 000 €	5%
4	– au-delà		10%

Dans les formules, on fera référence aux cellules \$B\$2 et \$B\$3 pour les valeurs des seuils, et aux cellules \$C\$2, \$C\$3 et \$C\$4 pour les valeurs des taux ; la feuille de calcul sera plus facile à maintenir en cas de variation de ces valeurs.

La fonction Somme.Si

Soit le tableau suivant :

	A	B	C	D
1	Désignation	Montant HT	TVA	
2			Taux	Montant
3	Alice Mutton	195,00 €	19,6%	38,22 €
4	Aniseed Syrup	50,00 €	5,5%	2,75 €
5	Boston Crab Meat	92,00 €	19,6%	18,03 €
6	Camembert Pierrot	170,00 €	5,5%	9,35 €
7	Carnarvon Tigers	312,50 €	5,5%	17,19 €
8				
9			TVA à 5,5%	29,29 €
10			TVA à 19,6%	56,25 €

La fonction SOMME.SI permet d'effectuer une somme conditionnelle.

Syntaxe : SOMME.SI(plage ; critère ; [somme plage])

Un exemple est plus parlant :

Total de la TVA à 5,5% : =SOMME.SI(C3:C7;5,5%;D3:D7)

Total de la TVA à 19,60% : =SOMME.SI(C3:C7;19,6%;D3:D7)

On fait la somme des valeurs de la colonne D quand le taux spécifié se trouve dans la colonne C.

La fonction SommeProd

Effectue la somme du produit des valeurs.

Soit le tableau suivant :

	A	B	C
1	Désignation	PU	Quantité
2	Alice Mutton	195,00 €	3
3	Aniseed Syrup	50,00 €	2
4	Boston Crab Meat	92,00 €	5
5	Camembert Pierrot	170,00 €	10
6	Carnarvon Tigers	312,50 €	4
7			
8	Montant total	4 095,00 €	

Montant total : =SOMMEPROD(B2:B6;C2:C6)